Algorithms

From the Algorithms Editor

The author attributions of the statistical algorithm groups 145-147 in issue 11 1 might seem to imply that the entire set was written by Mr Robert Wheeler. We would like to point out that only the functions PNORM and XNORM were in fact contributed by Mr Wheeler; they are part of a more comprehensive set from which we intend to publish more later.--NDT

ALGORITHM 149

Fourier Transformation of Observed Data by Harmonic Analysis

Glenn Schneider

Any periodic function Y=F(X) scaled to have a fundamental period of 2π may be represented by an infinite series of sine and cosine terms, called a Fourier series:

$$F(X) = \sum_{n=0}^{\infty} An \cos nX - Bn \sin nX$$

An and Bn are Fourier coefficients for F.

Numerous techniques are available for finding the Fourier coefficients of a function whose analytical form is known. In many practical applications, however, a relation between the dependent and independent variables is known only through observation. In such a case, there exists a finite number of known values X and Y. Fourier coefficients for this non-analytical function may be found, from observations evenly spaced through the period, by function *FT* which employs the method of harmonic analysis [1].

Input

The right argument *Y* must be a vector of an even number of observed independent values.

Output

The function returns an (N,2)-matrix, where *N*←(⍴*Y*)÷2. Column 0 contains the A-coefficients, Column 1 the B-coefficients. Coefficients in Row *I* relate to the terms of harmonic number *I*, 2 1○*I*×*X*.

Algorithm

```
    ∇ R←FT Y;M;N;⎕IO
[1]   ⍝ FOURIER ANALYSIS OF AN EVEN NUMBER
[2]   ⍝    OF EVENLY SPACED OBSERVATIONS Y.
[3]   ⍝ R'S COLUMNS ARE COEFFICIENTS A, B.
[4]   ⎕IO←0
[5]   M←(⍳2×N)×○÷N←(⍴Y)÷2
[6]   R← 1 2 ⍴⌽0,(+/Y)÷2×N
[7]   L1:R←R,[0](÷N)×+/(1 1 ∘.×Y)× 2 1 ∘.○
      (1↑⍴R)×M
[8]   →(N>¯1+1↑⍴R)/L1
[9]   R[N;]← .5 0 ×, ¯1 2 ↑R
    ∇
```

Example

```
      ⍝ Y = OBSERVATIONS AT REGULAR X

      Y
0 0.2618 0.7854 1.571 2.618 3.927 5.498
      7.33 9.425 11.78 14.4 17.28 20.42
      23.82 27.49 31.42 35.6 40.06 44.77
      49.74 54.98 60.48 66.24 72.26

      ⍝ THE FOURIER COEFFICIENTS ARE:

      ⎕←FCOEF←FT Y
 25.09       0
  4.411    ¯24.86
 ¯1.318    ¯12.21
 ¯2.379     ¯7.9
 ¯2.749     ¯5.668
 ¯2.919     ¯4.265
 ¯3.011     ¯3.272
 ¯3.065     ¯2.511
 ¯3.098     ¯1.889
 ¯3.119     ¯1.356
 ¯3.132     ¯0.8769
 ¯3.139     ¯0.4308
 ¯1.571      0

      ⍝ THEIR POWER SPECTRUM IS:

      1↓.5×+/FCOEF*2
318.7 75.45 34.04 19.84 13.36 9.887 7.848
      6.584 5.783 5.29 5.02 1.234
```

Comments

The data are assumed to represent a function continuous with its repeated cycles at the endpoints. Hence, care must be taken in determining where observations are to be truncated for transformation. If possible, the endpoints should represent dY/dX = 0. When this is not possible, the data may be folded before transformation: *Y*←*Y*,1↓⌽*Y*; this will result in suppression of spurious high-frequency components due to the finite length of the data and discontinuities at the endpoints.

The power spectrum of the data is given, except for Term 0, by .5×(+×/*FCOEF*)*2. The total averaged square power associated with the data is ((1 1↑*FCOEF*)*2)+.5×1↓+/*FCOEF**2.

Acknowledgment

We wish to thank Warner Computer Systems, Inc. of New York for their continuing computational support in astronomical research.

Glenn Schneider
Department of Astronomy
211 Space-Science Research Building
University of Florida
Gainesville, Florida
USA 32611

References

[1] H.S. Carslaw. Introduction to the theory of Fourier's series and integrals, Third revised edition, Dover (1930).

[2] D.C. Champeney. Fourier transforms and their physical applications, Academic Press (1973).

ALGORITHM 150

Inverse Fourier Transformation

Glenn Schneider

If the Fourier coefficients of a function Y = F(X) defined on the interval 0 to 2π are known (see preceding Algorithm), the function may be reconstructed from those coefficients [1]. In the case of a real finite series, the reconstructed function must equal the original function exactly at the regularly spaced finite values of X from which the Fourier coefficients were originally found. Given a finite Fourier series representing that function, it may be evaluated at point(s) X by the function *INVFT*.

Input

The right argument *R* is a two-column matrix, containing successive cosine-coefficients in the first column and sine-coefficients in the second. The left argument *X* is a vector of points at which the function is to be evaluated.

Output

The result is a vector of the values of the inverted function at the points *X*.

Algorithm

```
    ∇ F←X INVFT R;⎕IO
[1]   ⍝ INVERSE FOURIER TRANSFORM.
[2]   ⍝ R IS MATRIX OF COEFFICIENTS;
[3]   ⍝   X IS EVALUATION POINT(S) FOR
[4]   ⍝   FUNCTION REPRESENTED BY R.
[4]   ⎕IO←0
[5]   X← 2 0 1 ⍉ 2 1 ∘.○X×1↓⍳1↑⍴R
[6]   F←,(1 1 ↑R)++/+/((⍴X)⍴ 1 0 ↓R)×X
    ∇
```

Example

```
      ⍝ GIVEN A FINITE FOURIER REPRESENT'N:

      FCOEF
 25.09        0
  4.411     ¯24.86
 ¯1.318     ¯12.21
 ¯2.379      ¯7.9
 ¯2.749      ¯5.668
 ¯2.919      ¯4.265
 ¯3.011      ¯3.272
 ¯3.065      ¯2.511
 ¯3.098      ¯1.889
 ¯3.119      ¯1.356
 ¯3.132      ¯0.8769
 ¯3.139      ¯0.4308
 ¯1.571       0

      ⍝ CHOOSE 24 REGULARLY SPACED POINTS
      ⍝   ON 0 TO ○2. EVALUATE INVERSE AT
      ⍝    SEVERAL OF THESE POINTS:

      ⎕←X←(⍳24)×(○2)÷24
0 0.2618 0.5236 0.7854 1.047 1.309 1.571
      1.833 2.094 2.356 2.618 2.88 3.142
      3.403 3.665 3.927 4.189 4.451 4.712
      4.974 5.236 5.498 5.76 6.021

      X[0 1 2 23] INVFT FCOEF
¯2.132E¯13 0.2618  0.7854 72.26

      ⍝ EVALUATE AT SOME OTHER POINT
      0.85 INVFT FCOEF
3.814
```

Comment

If the original function represented by the coefficient matrix was defined on an interval other than 0≤*X*≤2, then the *X*-values should be normalized to this range before applying *INVFT* to them. For example, if the period of the original *X* is ¯2≤*X*≤2, the argument to *INVFT* should be 2×○(*X*+2)÷4.

Acknowledgment

We wish to thank Warner Computer Systems, Inc. of New York for their continuing computational support in astronomical research.

Glenn Schneider
Department of Astronomy
211 Space-Science Research Building
University of Florida
Gainesville, Florida
USA 32611

Reference

[1] W. Gellert et al. The VNR concise encyclopedia of mathematics, Van Nostrand Reinhold (1975).

⎕

ALGORITHM 151

Weighted Least Squares

Garry Helzer

The expression *X*←*B*⌹*A* defines *X* to be the solution in the least-squares sense of the linear system conventionally written:

(1) AX = B

Defining the vector of residuals by *R*←*B*-*A*+.×*X*, the above *X* minimizes the sum of squares *R*+.*2.

If some observations are more reliable than others, then one may wish to minimize instead (*W*×*R*)+.*2, where *W* is a vector of relative weights. This solution may be recomputed by rescaling the rows of the system (1) by the components of *W*. Thus the least-squares solution of (1) when weighted by *W* is:

```
(2)    X←(W×B)⌹ 1 1 2 ⍉ W∘.×A
```

For a first-degree (straight-line) fit to data points, *B* is the vector of measurements of the dependent variable, and *A* is constructed from the independent-variable values *C* by *A*←*C*∘.*0 1.

Comment

The outer product in (2) may fill the workspace. The following equivalent expression requires more interpretation time but can be more efficient:

```
(3)    X←(W×B)⌹(⍉(⌽⍴A)⍴W)×A
```

Garry Helzer
Department of Mathematics
University of Maryland
College Park, Maryland
USA 20740

⎕